



SOCIETY OF ACTUARIES

Article from:

CompAct

January 2008 – Issue 26

# The Microsoft Excel OFFSET Function

by *Damian A. Birnstihl*

**M**any of us use Microsoft Excel on a daily basis. We have mastered the basics, such as VLOOKUP and SUMPRODUCT. We can create charts and professional looking documents. We may even have ventured into more advanced features, like pivot tables and macros. But there are a number of functions and features of Excel that despite their power are used infrequently. One example is the OFFSET function. Learning and applying this function can significantly improve your efficiency in creating and using spreadsheets.

The OFFSET function returns the value in the cell a specified number of rows and columns away from a specified starting point. The syntax is **OFFSET (reference, rows, cols, height, width)**. The **reference** parameter is the starting point. Typically, it will be a single cell, but it can be a range of contiguous cells. The **rows** and **cols** parameters represent how many **rows** and columns away from the starting point you want to go. A positive value of the rows parameter means the target is below the reference cell, and a negative value means the target is above the reference cell. Similarly, a positive value of the **columns** parameter means the target is to the right of the reference cell, and a negative value means the target is to the left of the reference cell. The height and width parameters are optional; if omitted, Excel assumes the same height and width as **reference**.

For example, `OFFSET(B5,2,1)` returns the value two rows below and one column to the right of the reference cell B5, or in other words, the value in cell C7. `OFFSET(B5,-2,-1)` returns the value two rows above and one column to the left of cell B5, or in other words, the value in cell A3.

The real power of OFFSET is realized when it is used in conjunction with the ROW and COL-

UMN functions. The syntax of these functions is simple: `ROW(reference)` and `COLUMN(reference)`. For example, `ROW(F2)` returns a value of two, and `COLUMN(F2)` returns a value of six. If the reference parameter is omitted, the ROW returns the value of the row in which the formula is typed, and likewise for `COLUMN()`. For example, typing `=ROW()` in cell A3 returns a value of three, and typing `=COLUMN()` in the same cell returns a value of one.

Now let's look at an example that combines the use of OFFSET and ROW. Suppose you have a spreadsheet containing the months of the year in cells A2:A13, as shown in Figure 1.

Figure 1

	A
1	<b>Month</b>
2	January
3	February
4	March
5	April
6	May
7	June
8	July
9	August
10	September
11	October
12	November
13	December

Now suppose you want to reverse the data so that December is at the top of the list and January is at the bottom, with the results to be shown in cells C2:C13. No need to re-type the data or cut and paste; one simple formula can do the trick. In cell C2 type the formula `=OFFSET($A$13,2-ROW(),0)`. Then copy this formula to cells C3:C13, and you have the desired result as shown in Figure 2.



*Damian A. Birnstihl, FSA, MAAA is a director of actuarial services with Aetna. He can be reached at 602.659.1759 or at [Damian.Birnstihl@SchallerAnderson.com](mailto:Damian.Birnstihl@SchallerAnderson.com).*

Figure 2

	A	B	C
1	<b>Month</b>		
2	January		December
3	February		November
4	March		October
5	April		September
6	May		August
7	June		July
8	July		June
9	August		May
10	September		April
11	October		March
12	November		February
13	December		January

The key is figuring out how to set the parameters in the OFFSET function. This can be done using the following four-step process:

- First, determine what value you want in each target cell. In this example, we want cell C2 to have the value "December" from cell A13, C3 to have the value "November" from cell A12, etc.
- Second, choose a reference. This can be any cell on the spreadsheet, but a convenient choice is to use the cell in the original array that you want to appear in the first cell of the revised array. In this example that would be A13.
- Next, determine the offsets from this reference needed to produce the desired results. We stated above that we want C2 to contain the value from cell A13; this is zero rows below and zero columns to the right of the reference cell A13. In other words, C2 needs to evaluate to `OFFSET($A$13,0,0)`. Similarly, we want C3 to contain the value from A12; this is one row above and zero columns to the right of the reference cell A13. In other words, C3 needs to evaluate to `OFFSET($A$13,-1,0)`. Following the same logic, C4 must evaluate to `OFFSET($A$13,-2,0)`. For cells C5:C13 the emerging pattern holds: in each subsequent cell, the row offset decreases by one.

- Finally, look for a relationship between these offsets and the rows and columns of the cells in which the formulas will be entered, and translate the relationship into a formula. From step 3 we know that the row offset must be 0 in cell C2, -1 in cell C3, -2 in cell C4, etc. Notice that each row offset can be expressed as `2-ROW()`. The column offset in this example is always zero. Thus, the desired formula is `OFFSET($A$13,2-ROW(),0)`.

The process of determining the appropriate parameters becomes second nature with a little practice.

A word of caution is in order. The formula in the preceding example may no longer produce the desired results if rows are inserted into the spreadsheet. This can be seen by inserting a row at the top of the spreadsheet. Ideally, this action would not cause our results to change. In fact, inserting a row does indeed cause our resulting list to start at November instead of December. However, this can easily be avoided. In the example, we used a row offset of `2-ROW()`. The fixed value two was appropriate before the row was inserted, but is not appropriate afterwards. So instead of a fixed value, we can reference the cell itself. If you're working the example on your computer, delete the row that you added, and in place of the original formula in cell C2, type in `=OFFSET($A$13,ROW($C$2)-ROW(),0)` and copy this formula to cells C3:C13. Now that the fixed reference has been replaced with a ROW reference, inserting rows at the top of the spreadsheet will not affect the results. Note that I have "anchored" both `$A$13` and `$C$2` using dollar signs.

Now, let's look at a similar example using OFFSET and COLUMN. Suppose you have the spreadsheet shown in Figure 3.

**... there are a number of functions and features of Excel that despite their power are used infrequently.**

*(continued on page 6)*

Figure 3

	A	B	C	D	E
1	Alpha	Bravo	Charlie	Delta	Echo

Our task is to reverse the list and put the results in cells A3:E3. In cell A3, type =OFFSET(\$E\$1,0,COLUMN(\$A\$3)-COLUMN()), and then copy this formula to cells B3:E3. This produces the desired result as shown in Figure 4.

Figure 4

	A	B	C	D	E
1	Alpha	Bravo	Charlie	Delta	Echo
2					
3	Echo	Delta	Charlie	Bravo	Alpha

Next, let's take the original list in Figure 3 and convert it from a 1x5 array into a 5x1 array, or in other words convert the row of data into a column with Alpha at the top and Echo at the bottom, and let's put the results in cells G1:G5. We'll use the four-step process introduced earlier to guide us:

- We want cell G1 to contain the value "Alpha" from cell A1, G2 to contain "Bravo" from B1, etc.
- We choose \$A\$1 as the reference since it contains the desired value in the first target cell, G1.
- G1 must evaluate to OFFSET(\$A\$1,0,0), G2 must evaluate to OFFSET(\$A\$1,0,1), G3 must evaluate to OFFSET(\$A\$1,0,2), etc.
- We see that the row offset is always zero and that the column offset is the current row minus one. Thus, the formula to type into G1 and to copy to G2:G5 is =OFFSET(\$A\$1,0,ROW()-ROW(\$G\$1)).

Now that we've looked at some basic examples, let's look at some more advanced applications of OFFSET. Suppose we have the 3x5 array shown in Figure 5, and suppose that we want to transpose it to a 5x3 array and put the result in cells G1:I5.

Figure 5

	A	B	C	D	E
1	10	40	70	100	130
2	20	50	80	110	140
3	30	60	90	120	150

Using the four-step process:

- We want the first row of the original array to become the first column of the resulting array, and likewise for the second and third rows of the original array.
- A convenient choice for the reference is cell \$A\$1.
- G1 must evaluate to OFFSET(\$A\$1,0,0). We want cell H1 to contain the value from cell A2, so H1 must evaluate to OFFSET(\$A\$1,1,0). Similarly, we want cell G2 to contain the value from cell B1, so G2 must evaluate to OFFSET(\$A\$1,0,1). If necessary, look at some additional cells until the pattern becomes evident.
- From the previous step, we note that the row offset relates to the column in the transposed array and that the column offset relates to the row in the transposed array. The formula is =OFFSET(\$A\$1,COLUMN()-COLUMN(\$G\$1),ROW()-ROW(\$G\$1)). Type this formula in cell G1 and then copy the formula to cells G1:I5 to obtain the desired result as shown in Figure 6.

Figure 6

	G	H	I
1	10	20	30
2	40	50	60
3	70	80	90
4	100	110	120
5	130	140	150

I hope that you are beginning to see the power of the OFFSET function. But it can do more than just flip arrays. For our final example, we will look at a practical example of how OFFSET can be used along with SUM. Consider the hypothetical claim triangle in Figure 7.

Figure 7

	A	B	C	D	E
1		July	August	September	October
2	July	10			
3	August	20	30		
4	September	40	50	60	
5	October	70	80	90	100

Here, each column represents an incurred month and each row represents a month of payment. Now suppose that we want to build a table of cumulative paid claims by

incurred month as shown in Figure 8. This table can be built with a single formula!

Figure 8

	G	H	I	J	K
1	Incurring Month	Lag 0	Lag 1	Lag 2	Lag 3
2	July	10	30	70	140
3	August	30	80	160	
4	September	60	150		
5	October	100			

For example, cell J3 represents claims incurred in August and paid through lag 2, i.e., October. The value is 30 + 50 + 80 = 160. Figure 9 shows the calculations necessary to produce the desired results. This is just an intermediate step that we will use as a guide to determine the appropriate OFFSET formula.

Figure 9

	G	H	I	J	K
1		Lag 0	Lag 1	Lag 2	Lag 3
2	July	=B2	=B2+B3	=B2+B3+B4	=B2+B3+B4+B5
3	August	=C3	=C3+C4	=C3+C4+C5	
4	September	=D4	=D4+D5		
5	October	=E5			

So, in each case we want to sum starting at an offset, an equal number of rows and columns from cell B2, with both the row and column offset determined by the row within the array. Further, each summation is over an nx1 range, where n is determined by the column within the array. We also note that the lower right half of the array is blank because these represent payment months in the

future. We can even make our formula account for this by using an IF statement. Here is the formula:

```
=IF(ROW()-ROW($H$2)+COLUMN()-COLUMN($H$2)>3,"",SUM(OFFSET($B$2,ROW()-ROW($H$2),ROW()-COLUMN($H$2)+1,1)))
```

Typing this formula into cell H2 and copying it to cells H2:K5 produces the desired result. Note that this formula uses the optional height and width parameters in the OFFSET function. Let's evaluate the formula in cell J3 to see how it works. In J3 the formula is evaluated as follows:

```
=IF(3-2+10-8>3,"",SUM(OFFSET($B$2,3-2,3-2,10-8+1,1)))
```

```
=IF(3>3,"",SUM(OFFSET($B$2,1,1,3,1)))
```

```
=SUM(C3:C5)
```

```
=160
```

As you can see, the OFFSET function is both powerful and efficient for manipulating arrays of data. We have looked at several examples of how OFFSET can be used along with ROW, COLUMN, and SUM to reverse, transpose, and sum arrays with a single formula. Master the use of OFFSET and you may never have to cut and paste again! 📄